

University of Huddersfield

VHDL Circuit Modelling and Simulation in Intel's Quartus Prime software

Digital System Integration


Jack Parkinson U1552044
11-11-2018

Contents

1.	7/4 Decoder	2
2.1.	Design.....	2
2.2.	VHDL Code.....	4
2.3.	ModelSim Waveform	6
2.	15/11 Decoder	9
3.1.	Design.....	9
3.2.	VHDL Code.....	11
3.3.	ModelSim Waveform	13
3.	7/4 Encoder.....	17
4.1.	Design.....	17
4.2.	VHDL Code.....	18
4.3.	ModelSim Waveform	20
4.	7/4 Linear Sequential Coding System	23
5.1.	Design.....	23
5.2.	VHDL Code.....	24
5.3.	ModelSim Waveform	26
5.	Generic Decoder.....	28
6.1.	Design.....	28
6.2.	VHDL Code.....	30
6.3.	ModelSim Waveform	33
6.	Generic Encoder	36
7.1.	Design.....	36
7.2.	VHDL Code.....	38
7.3.	ModelSim Waveform	40
7.	Generic Linear Sequential Coding System	42
8.1.	Design.....	42
8.2.	VHDL Code.....	43
8.3.	ModelSim Waveform	45
8.	Parallel-In Encoder	47
9.1.	Design.....	47
9.2.	VHDL Code.....	49
9.3.	ModelSim Waveform	51

1. 7/4 Decoder

2.1. Design



The diagram above illustrates the operation of the 7/4 decoder which should be:

1. Can be reset at any time by setting the res port high
2. Decodes messages sent in serially on the sin port in packets of 7-bits
3. Outputs the 4-bits of data contained with the 7-bit message in parallel with any single bit errors corrected
4. Continues to operate once the 7-bit message has been decoded i.e. two or more 7-bit messages can be sent in one after another and be successfully decoded without having to reset the decoder

The purpose of each component illustrated in the diagram above is:

- **sin input port** – receives input signal
- **clk input port** - receives clock signal
- **res input port** - receives reset signal
- **dout output port** – transmits error corrected signal
- **⊕ logic** – decodes input signal
- **control shift register** – keeps track of the number of bits decoded. When MSB is “1” it resets the rxlss register to “000” and it initiates the error check and assessor/corrector processes. Setting the rxlss register to “000” allows the next 7-bits to be correctly decoded
- **rxlss shift register** – stores the last 3 decoded bits to decode the input signal
- **sipo shift register** – stores the last 7 decoded bits (complete message) for the error check and assessor/corrector processes

The Boolean expression for the logic which decodes the input signal is:

$$sipo(6) \& rxlss(0) = sin \oplus rxlss(0) \oplus rxlss(2)$$

This gives the impulse response:

<i>sin</i>	<i>rxlss(0)</i>	<i>rxlss(1)</i>	<i>rxlss(2)</i>	<i>sipo(6) & rxlss(0)</i>
1	0	0	0	1
0	1	0	0	1
0	1	1	0	1
0	1	1	1	0
0	0	1	1	1
0	1	0	1	0
0	0	1	0	0

From this impulse response, the Noise Look-Up table (LUT) can be given as:

<i>sin</i>	<i>sipo</i>						
	6	5	4	3	2	1	0
0000001	0	0	1	0	1	1	1
0000010	0	1	0	1	1	1	0
0000100	1	0	1	1	1	0	0
0001000	0	1	1	1	0	0	0
0010000	1	1	1	0	0	0	0
0100000	1	1	0	0	0	0	0
1000000	1	0	0	0	0	0	0

- error code
- correction code
- ignored (data = "0000")


2.2. VHDL Code

```


1  library IEEE; -- include ieee library
2  use IEEE.STD_LOGIC_1164.all; -- include STD_LOGIC_1164 package from IEEE library to use the std_logic and
3  -- std_logic_vector data types which adds U (undefined) and Z (high impedance) assignments to the
4  -- standard VHDL bit and bit_vector data types
5  use IEEE.STD_LOGIC_UNSIGNED.all; -- Include STD_LOGIC_UNSIGNED package from IEEE library to be able to perform arithmetic,
6  -- conversion and comparison operations on the std_logic_vector data type
7
8  -- define the interface between the 7/4 decoder and its external environment
9  ENTITY decoder74 IS
10   PORT (
11     clk, res, sin : IN STD_LOGIC; -- define the single point input ports (clock, reset & serial-in)
12     dout           : OUT STD_LOGIC_VECTOR(3 downto 0) -- define the 4 point output port (data-out)
13   );
14 END decoder74;
15
16 -- define the internal organisation and operation of the 7/4 decoder
17 ARCHITECTURE rtl OF decoder74 IS
18   -- architecture declarations
19   SIGNAL control : STD_LOGIC_VECTOR(6 downto 0); -- define the 7-bit control shift register
20   SIGNAL rxlss   : STD_LOGIC_VECTOR(2 downto 0); -- define the 3-bit rxlss (receive-linear-sequential-system) shift register
21   SIGNAL sipo    : STD_LOGIC_VECTOR(6 downto 0); -- define the 7-bit sipo (serial-in-parallel-out) shift register
22
23   -- concurrent statements
24 BEGIN
25   -- define all linear sequential logic as a single process
26   PROCESS BEGIN
27     WAIT UNTIL RISING_EDGE (clk); -- ensures that each line of code in this process is dependent on a rising clock edge
28
29     -- when the reset port is set high, reset all registers to known state
30     IF (res = '1') THEN
31       control <= "1000000";
32       rxlss <= "000";
33       sipo <= "0000000";
34     -- when the reset port is set low, decode data coming in on the sin port sequentially and correct single bit errors
35     ELSIF (res = '0') THEN
36       rxlss(0) <= (sin XOR rxlss(2) XOR rxlss(0)) AND NOT(control(0)); -- decode the data coming in on the sin port by
37       -- assigning the LSB of the rxlss register to the
38       -- sin port XOR'ed with the MSB and LSB of the
39       -- rxlss register. However, when all 7-bits
40       -- have been received on the sin port (MSB of the
41       -- control register is low) set the LSB of the
42       -- rxlss register low
43
44       -- shift the current value of each bit in the rxlss register to the next MSB but set them low when all 7-bits
45       -- have been received
46       rxlss(1) <= rxlss(0) AND NOT(control(0));
47       rxlss(2) <= rxlss(1) AND NOT(control(0));
48
49       -- store decoded data in sipo register by assigning the MSB to the sin port XOR'ed with the MSB and LSB of the
50       -- rxlss register and shifting the current value of each bit down to the next LSB
51       sipo <= (sin XOR rxlss(2) XOR rxlss(0)) & sipo(6 downto 1);
52
53       -- keep track of the number of bits received sequentially on the sin port by shifting each bit to the next LSB and
54       -- looping the LSB back round to the MSB (a single bit will always be high while the others are low)
55       control <= (control(0)) & (control(6 downto 1));
56
57       -- when the sipo register has been filled (all 7-bits have been received on sin port and decoded) check for single
58       -- bit errors and correct them by comparing the 3 MSB's of the sipo register (error bits) with the noise look-up
59       -- table
60       IF (control(6) = '1') THEN
61         CASE (sipo(6 downto 4)) IS
62           -- when single bit error is present XOR data bits of sipo with appropriate value from look-up table and send
63           -- out the corrected data in parallel on the dout port
64           WHEN "001" => dout <= (sipo(3 downto 0)) XOR ("0111");
65           WHEN "010" => dout <= (sipo(3 downto 0)) XOR ("1110");
66           WHEN "101" => dout <= (sipo(3 downto 0)) XOR ("1100");
67           WHEN "011" => dout <= (sipo(3 downto 0)) XOR ("1000");
68           -- when no single bit error is present or the data bits of sipo are "000" send out the data bits of sipo in
69           -- parallel on the dout port
70           WHEN OTHERS => dout <= sipo(3 downto 0);
71         END CASE;
72       END IF;
73     END IF;
74   END PROCESS;
75 END rtl;

```

Compiled to schematic




2.3. ModelSim Waveform



1. Reset registers to known state
2. Serially load in "0000001"
3. After 7-bits have been decoded (control = "1000000"), sipo = "0010111" which matches the impulse response on page 3

dout = "0000" throughout as sipo contains the error code "001" from the LUT which means the data bits from sipo (four LSB's) are XOR'ed with the corresponding correction code from the LUT

$$\begin{array}{l} \text{sipo} \quad \text{code} \quad \text{dout} \\ 0111 \oplus 0111 = 0000 \end{array}$$



1. Reset registers to known state
2. Serially load in "0111011"
3. After 7-bits have been decoded (control = "1000000"), sipo = "0010001" to check if this is correct, XOR corresponding rows from LUT


sin = 0 1 1 1 0 1 1
 ↓ ↓ ↓ ↓ ↓ ↓ ↓
 0010111
 0101110
 0111000
 1110000
 1100000
 sipo = 0010001

4. Serially load in "1100101"

When control = "1000000", sipo = "0010001" which contains the error code "001" from LUT. To work out what dout for the first 7-bit message load in during **step 2**, the data bits from sipo need to be XOR'ed with the corresponding correction code from the LUT


sipo code dout
 0001 ⊕ 0111 = 0110

5. After 7-bits have been decoded (control = "1000000"), sipo = "1101011" to check if this is correct, XOR corresponding rows from LUT



2. 15/11 Decoder

3.1. Design



The diagram above illustrates the operation of the 15/11 decoder which should be:

1. Can be reset at any time by setting the res port high
2. Decodes messages sent in serially on the sin port in packets of 15-bits
3. Outputs the 11-bits of data contained with the 15-bit message in parallel with any single bit errors corrected
4. Continues to operate once the 15-bit message has been decoded i.e. two or more 15-bit messages can be sent in one after another and be successfully decoded without having to reset the decoder

The purpose of each component illustrated in the diagram above is:

- **sin input port** – receives input signal
- **clk input port** - receives clock signal
- **res input port** - receives reset signal
- **dout output port** – transmits error corrected signal
- **⊕ logic** – decodes input signal
- **control shift register** – keeps track of the number of bits decoded. When MSB is “1” it resets the rxlss register to “0000” and it initiates the error check and assessor/corrector processes. Setting the rxlss register to “0000” allows the next 15-bits to be correctly decoded
- **rxlss shift register** – stores the last 4 decoded bits to decode the input signal
- **sipo shift register** – stores the last 15 decoded bits (complete message) for the error check and assessor/corrector processes

The Boolean expression for the logic which decodes the input signal is:

$$sipo(14) \& rxlss(0) = sin \oplus rxlss(2) \oplus rxlss(3)$$

This gives the impulse response:

<i>sin</i>	<i>rxlss(0)</i>	<i>rxlss(1)</i>	<i>rxlss(2)</i>	<i>rxlss(3)</i>	<i>sipo(14) & rxlss(0)</i>
1	0	0	0	0	1
0	1	0	0	0	0
0	0	1	0	0	0
0	0	0	1	0	1
0	1	0	0	1	1
0	1	1	0	0	0
0	0	1	1	0	1
0	1	0	1	1	0
0	0	1	0	1	1
0	1	0	1	0	1
0	1	1	0	1	1
0	1	1	1	0	1
0	1	1	1	1	0
0	0	1	1	1	0
0	0	0	1	1	0

From this impulse response, the Noise Look-Up table (**LUT**) can be given as:

<i>sin</i>	<i>sipo</i>														
	14	13	12	11	10	9	8	7	6	5	4	3	2	1	0
0000000000000001	0	0	0	1	1	1	0	1	0	1	1	0	0	1	
0000000000000010	0	0	1	1	1	0	1	0	1	1	0	0	1	0	
0000000000000100	0	1	1	1	1	0	1	0	1	1	0	0	1	0	
0000000000001000	1	1	1	1	0	1	0	1	1	0	0	1	0	0	
0000000000100000	1	1	1	0	1	0	1	1	0	0	1	0	0	0	
0000000001000000	1	1	0	1	0	1	1	0	0	1	0	0	0	0	
0000000010000000	1	0	1	0	1	1	0	0	1	0	0	0	0	0	
0000000100000000	0	1	0	1	1	0	0	1	0	0	0	0	0	0	
0000001000000000	1	0	1	1	0	0	1	0	0	0	0	0	0	0	
0000010000000000	0	1	1	0	0	1	0	0	0	0	0	0	0	0	
0000100000000000	1	1	0	0	1	0	0	0	0	0	0	0	0	0	
0001000000000000	1	0	0	1	0	0	0	0	0	0	0	0	0	0	
0010000000000000	0	0	1	0	0	0	0	0	0	0	0	0	0	0	
0100000000000000	0	1	0	0	0	0	0	0	0	0	0	0	0	0	
1000000000000000	1	0	0	0	0	0	0	0	0	0	0	0	0	0	

- error code
- correction code
- ignored (data = "000000000000")


3.2. VHDL Code

```


1  library IEEE; -- include ieee library
2  use IEEE.STD_LOGIC_1164.all; -- include STD_LOGIC_1164 package from IEEE library to use the std_logic and
3  -- std_logic_vector data types which adds U (undefined) and Z (high impedance) assignments to the
4  -- standard VHDL bit and bit_vector data types
5  use IEEE.STD_LOGIC_UNSIGNED.all; -- Include STD_LOGIC_UNSIGNED package from IEEE library to be able to perform arithmetic,
6  -- conversion and comparison operations on the std_logic_vector data type
7
8  -- define the interface between the 15/11 decoder and its external environment
9  ENTITY decoder1511 IS
10     PORT (
11         clk, res, sin : IN STD_LOGIC; -- define the single point input ports (clock, reset & serial-in)
12         dout           : OUT STD_LOGIC_VECTOR(10 downto 0) -- define the 11 point output port (data-out)
13     );
14 END decoder1511;
15
16 -- define the internal organisation and operation of the 15/11 decoder
17 ARCHITECTURE behaviour OF decoder1511 IS
18     -- architecture declarations
19     SIGNAL control : STD_LOGIC_VECTOR(14 downto 0); -- define the 15-bit control shift register
20     SIGNAL rxlss   : STD_LOGIC_VECTOR(3 downto 0); -- define the 4-bit rxlss (receive-linear-sequential-system) shift register
21     SIGNAL sipo    : STD_LOGIC_VECTOR(14 downto 0); -- define the 15-bit sipo (serial-in-parallel-out) shift register
22
23     -- concurrent statements
24 BEGIN
25     -- define all linear sequential logic as a single process
26     PROCESS BEGIN
27         WAIT UNTIL RISING_EDGE (clk); -- ensures that each line of code in this process is dependent on a rising clock edge
28
29         -- when the reset port is set high, reset all registers to known state
30         IF (res = '1') THEN
31             control <= "100000000000000";
32             rxlss <= "0000";
33             sipo <= "000000000000000";
34         -- when the reset port is set low, decode data coming in on the sin port sequentially and correct single bit errors
35         ELSIF (res = '0') THEN
36             rxlss(0) <= (sin XOR rxlss(3) XOR rxlss(2)) AND NOT(control(0)); -- decode the data coming in on the sin port by
37                                     -- assigning the LSB of the rxlss register to the
38                                     -- sin port XOR'ed with the two MSB's of the
39                                     -- rxlss register. However, when all 15-bits
40                                     -- control register is low) set the LSB of the
41                                     -- rxlss register low
42
43             -- shift the current value of each bit in the rxlss register to the next MSB but set them low when all 15-bits
44             -- have been received
45             rxlss(1) <= rxlss(0) AND NOT(control(0));
46             rxlss(2) <= rxlss(1) AND NOT(control(0));
47             rxlss(3) <= rxlss(2) AND NOT(control(0));
48
49             -- store decoded data in sipo register by assigning the MSB to the sin port XOR'ed with the two MSB's of the
50             -- rxlss register and shifting it down
51             sipo <= (sin XOR rxlss(3) XOR rxlss(2)) & sipo(14 downto 1);
52
53             -- keep track of the number of bits received sequentially on the sin port by shifting each bit to the next LSB and
54             -- looping the LSB back round to the MSB (a single bit will always be high while the others are low)
55             control <= (control(0)) & (control(14 downto 1));
56
57             -- when the sipo register has been filled (all 15-bits have been received on sin port and decoded) check for single
58             -- bit errors and correct them by comparing the 4 MSB's of the sipo register (error bits) with the noise look-up
59             -- table
60             IF (control(14) = '1') THEN
61                 CASE (sipo(14 downto 11)) IS
62                     -- when single bit error is present XOR data bits of sipo with appropriate value from look-up table and send
63                     -- out the corrected data in parallel on the dout port
64                     WHEN "0001" => dout <= (sipo(10 downto 0)) XOR ("11101011001");
65                     WHEN "0011" => dout <= (sipo(10 downto 0)) XOR ("11010110010");
66                     WHEN "0111" => dout <= (sipo(10 downto 0)) XOR ("10101100100");
67                     WHEN "1111" => dout <= (sipo(10 downto 0)) XOR ("01011001000");
68                     WHEN "1110" => dout <= (sipo(10 downto 0)) XOR ("10110010000");
69                     WHEN "1101" => dout <= (sipo(10 downto 0)) XOR ("01100100000");
70                     WHEN "1010" => dout <= (sipo(10 downto 0)) XOR ("11001000000");
71                     WHEN "0101" => dout <= (sipo(10 downto 0)) XOR ("10010000000");
72                     WHEN "1011" => dout <= (sipo(10 downto 0)) XOR ("00100000000");
73                     WHEN "0110" => dout <= (sipo(10 downto 0)) XOR ("01000000000");
74                     WHEN "1100" => dout <= (sipo(10 downto 0)) XOR ("10000000000");
75                     -- when no single bit error is present or the data bits of sipo are "0000" send out the data bits of sipo in
76                     -- parallel on the dout port
77                     WHEN OTHERS => dout <= sipo(10 downto 0);
78                 END CASE;
79             END IF;
80         END IF;
81     END PROCESS;
82 END behaviour;

```

Compiled to schematic



3.3. ModelSim Waveform





1. Reset registers to known state
2. Serially load in "0000000000000001"

3. After 15-bits have been decoded (control = “100000000000000”), sipo = “000111101011001” which matches the impulse response on page 10


dout = “00000000000” throughout as sipo contains the error code “0001” from the LUT which means the data bits from sipo (eleven LSB’s) are XOR’ed with the corresponding correction code from the LUT

$$\begin{array}{ccc} \textit{sipo} & \textit{code} & \textit{dout} \\ 11101011001 \oplus 11101011001 & = & 00000000000 \end{array}$$



1. Reset registers to known state
2. Serially load in "10000000001110"

- After 15-bits have been decoded (control = "100000000000000"), sipo = "001100100011110" to check if this is correct, XOR corresponding rows from LUT




- Serially load in "000000000000001"

When control = "100000000000000", sipo = "001100100011110" which contains the error code "0011" from LUT. To work out what dout for the 15-bit message load in during **step 2**, the data bits from sipo need to be XOR'ed with the corresponding correction code from the LUT

$$\begin{array}{ccc}
 \text{sipo} & \text{code} & \text{dout} \\
 00110011110 \oplus 11010110010 & = & 11110101100
 \end{array}$$

3. 7/4 Encoder

4.1. Design



The diagram above illustrates the operation of the 7/4 encoder which should be:

1. Can be reset at any time by setting the res port high
2. Encodes messages sent in serially on the sin port in packets of 7-bits
3. Outputs the encoded messages serially
4. Continues to operate once the 7-bit message has been encoded i.e. two or more 7-bit messages can be sent in one after another and be successfully encoded without having to reset the encoder

The purpose of each component illustrated in the diagram above is:

- **sin input port** – receives input signal
- **clk input port** - receives clock signal
- **res input port** - receives reset signal
- **sout output port** – transmits encoded signal
- **⊕ logic** – encodes input signal
- **rxlss shift register** – stores the last 3 bits input to encode the input signal

The Boolean expression for the logic which encodes the input signal is:

$$sout = sin \oplus rxlss(0) \oplus rxlss(2)$$

This gives the impulse response:

<i>sin</i>	<i>rxlss(0)</i>	<i>rxlss(1)</i>	<i>rxlss(2)</i>	<i>sout</i>
1	0	0	0	1
0	1	0	0	1
0	0	1	0	0
0	0	0	1	1
0	0	0	0	0
0	0	0	0	0
0	0	0	0	0

From this impulse response, the Noise Look-Up table (**LUT**) can be given as:

<i>sin</i>	<i>sout</i>
0000001	0001011
0000010	0010110
0000100	0101100
0001000	1011000
0010000	0110000
0100000	1100000
1000000	1000000


4.2. VHDL Code

```


1  library IEEE; -- include ieee library
2  use IEEE.STD_LOGIC_1164.all; -- include STD_LOGIC_1164 package from IEEE library to use the std_logic and
3  -- std_logic_vector data types which adds U (undefined) and Z (high impedance) assignments to the
4  -- standard VHDL bit and bit_vector data types
5  use IEEE.STD_LOGIC_UNSIGNED.all; -- Include STD_LOGIC_UNSIGNED package from IEEE library to be able to perform arithmetic,
6  -- conversion and comparison operations on the std_logic_vector data type
7
8  -- define the interface between the 7/4 encoder and its external environment
9  ENTITY encoder74 IS
10   PORT (
11     clk, res, sin : IN STD_LOGIC; -- define the single point input ports (clock, reset & serial-in)
12     sout          : OUT STD_LOGIC -- define the single point output port (serial-out)
13   );
14   END encoder74;
15
16  -- define the internal organisation and operation of the 7/4 encoder
17  ARCHITECTURE rtl OF encoder74 IS
18   -- architecture declarations
19   SIGNAL rxlss : STD_LOGIC_VECTOR(0 to 2); -- define the 3-bit rxlss (receive-linear-sequential-system) shift register
20
21   -- concurrent statements
22   BEGIN
23     -- combination logic to send encoded data out on the sout port by XOR'ing the sin port with the MSB and LSB of the rxlss
24     -- register
25     sout <= sin XOR rxlss(0) XOR rxlss(2);
26
27     -- define all linear sequential logic as a single process
28     PROCESS BEGIN
29       WAIT UNTIL RISING_EDGE (clk); -- ensures that each line of code in this process is dependent on a rising clock edge
30
31       -- when the reset port is set high, reset register to known state
32       IF (res = '1') THEN
33         rxlss <= "000";
34       -- when the reset port is set low, assign the LSB of the rxlss register to the sin port and shift it to the next MSB
35       -- of the rxlss register
36       ELSIF (res = '0') THEN
37         rxlss <= sin & rxlss(0 to 1);
38       END IF;
39     END PROCESS;
40   END rtl;

```

Compiled to schematic





4.3. ModelSim Waveform



1. Reset registers to known state
2. Serially load in "0000001" and send encoded signal out

At first glance, sout does not appear to match that of the impulse response on page 18 due to the encoding logic being implemented as combination logic rather than sequential logic so it is not depend on the clock. Due to the delay in the combination logic, the decoder will read falling edges on sout as “1” and rising edges as “0” hence sout does match the impulse response “0001011”





1. Reset registers to known state
3. Serially load in “0001101” and send encoded signal out

To check what sout should be, XOR corresponding rows from LUT

sin = 0 0 0 1 1 0 1


$$\begin{array}{r}
 \begin{array}{c} \text{0001011} \\ \text{0101100} \\ \text{1011000} \end{array} \\
 \hline
 \text{sout} = 1111111
 \end{array}$$

4. Serially load in “0000011” and send encoded signal out

To check what sout should be, XOR corresponding rows from LUT


sin = 0 0 0 0 0 1 1

$$\begin{array}{r}
 \begin{array}{c} \text{0001011} \\ \text{0010110} \end{array} \\
 \hline
 \text{sout} = 0011101
 \end{array}$$



4. 7/4 Linear Sequential Coding System

5.1. Design



The diagram above illustrates the operation of the 7/4 LSS which should be:

1. Can be reset at any time by setting the res port high
2. Encodes and decodes messages sent in serially on the sin port in packets of 7-bits such that any single bit errors introduced in the transmission line between the encoder and decoder are corrected
3. Outputs the 4-bits of data contained with the 7-bit message in parallel
4. Continues to operate once the 7-bit message has been encoded and decoded i.e. two or more 7-bit messages can be sent in one after another and be successfully encoded and decoded without having to reset the LSS

The purpose of each component illustrated in the diagram above is:

- **sin input port** – receives input signal
- **clk input port** - receives clock signal
- **res input port** - receives reset signal
- **test input port & \oplus logic** – inverts the signal present on transmission line between the encoder and decoder to introduce errors
- **dout output port** – transmits error corrected signal


5.2. VHDL Code

```


1  library IEEE; -- include ieee library
2  use IEEE.STD_LOGIC_1164.all; -- include STD_LOGIC_1164 package from IEEE library to use the std_logic and
3                                -- std_logic_vector data types which adds u (undefined) and z (high impedance) assignments to the
4                                -- standard VHDL bit and bit_vector data types
5  use IEEE.STD_LOGIC_UNSIGNED.all; -- Include STD_LOGIC_UNSIGNED package from IEEE library to be able to perform arithmetic,
6                                -- conversion and comparison operations on the std_logic_vector data type
7
8  -- define the interface between the 7/4 linear sequential coding system (lss) and its external environment
9  ENTITY lss74 IS
10     PORT (
11         clk, res, sin, test : IN STD_LOGIC; -- define the single point input ports (clock, reset, serial-in & test)
12         dout : OUT STD_LOGIC_VECTOR(3 downto 0) -- define the 4 point output port (data-out)
13     );
14 END lss74;
15
16 -- define the internal organisation and operation of the 7/4 lss
17 ARCHITECTURE rtl OF lss74 IS
18     -- architecture declarations
19     SIGNAL t1 : STD_LOGIC; -- define the serial transmission line between the output of the encoder and input of the decoder
20
21     -- concurrent statements
22 BEGIN
23     -- combination logic to instantiate the 7/4 encoder and 7/4 decoder entities inside the 7/4 lss
24     encoder74 : ENTITY work.encoder74(rtl) PORT MAP (
25         sin => sin, -- connect the sin port of the 7/4 encoder directly to the sin port of the 7/4 lss
26         clk => clk, -- connect the clk port of the 7/4 encoder directly to the clk port of the 7/4 lss
27         res => res, -- connect the res port of the 7/4 encoder directly to the res port of the 7/4 lss
28         sout => t1 -- connect the sin port of the 7/4 encoder to the transmission line
29     );
30
31     decoder74 : ENTITY work.decoder74(rtl) PORT MAP (
32         sin => t1 XOR test, -- connect the sin port of the 7/4 decoder to the transmission line XOR'ed with the test port
33                             -- of the 7/4 lss. This allows errors to be introduced as the signal coming out of the 7/4
34                             -- encoder will be inverted when the test port is set high
35         clk => clk, -- connect the clk port of the 7/4 decoder directly to the clk port of the 7/4 lss
36         res => res, -- connect the res port of the 7/4 decoder directly to the res port of the 7/4 lss
37         dout => dout -- connect the dout port of the 7/4 decoder directly to the dout port of the 7/4 lss
38     );
39 END rtl;

```

Compiled to schematic



5.3. ModelSim Waveform




1. Reset registers to known state
2. Serially load in "0001011" and introduce single bit error
3. Serially load in "0000101" and introduce single bit error

`dout` = 1101 which matches the data bits (first 4 bits) of the 7-bit message load in during **step 2** hence the single bit error was corrected successfully

4. `dout` = 0101 which matches the data bits of the 7-bit message load in during **step 3** hence the single bit error was corrected successfully

To prove the test signal is defiantly introducing an error, the next waveform shows what happens when a multiple bit error is introduced with the same input data



1. Reset registers to known state
2. Serially load in "0001011" and introduce single bit error
3. Serially load in "0000101" and introduce single bit error


dout = 1001 which doesn't match the data bits of the 7-bit message load in during step 2 hence the test pin is introducing errors correctly

4. dout = 1101 which doesn't match the data bits of the 7-bit message load in during step 3 hence test pin is introducing errors correctly

5. Generic Decoder

6.1. Design

Key			
■	Entity	n	Error Bits
■	Port	d	Data Bits ($2^{n-(n+1)}$)
■	Sequential Logic (dependent on clk)	a	All Bits ($n + 2^{n-(n+1)}$)
■	Register		



The diagram above illustrates the operation of the generic decoder which should be the same as that of the decoders described in **sections 1** and **2**. The difference being that the VHDL code for the generic decoder can be changed from a 7/4 to a 15/11 decoder by changing a single generic value. In this case, the generic value that is altered is the number of error bits used. For a 7/4 decoder this would be 3 and for a 15/11 it would be 4. Error bits was chosen over other generic values as it ensures the coding system is perfect.

The Boolean expression for the logic which decodes the input signal is:

$$sipo(a - 1) \& rxlss(0) = sin \oplus rxlss(n - 2) \oplus rxlss(n - 1)$$

Where:

n = number of error bits

a = number of bits in a complete message ($n + 2^{n-(n+1)}$)

When the number of error bits is set to 4, the impulse response and LUT is identically to that of the 15/11 decoder described in **section 2** on **page 9**. However, when the number of error bits is set to 3 the impulse response and LUT differs from that of the 7/4 decoder describe in **section 1** on **page 3** as the generic decoder always XOR's the two MSB's of rxlss rather than the MSB and LSB. This was done to keep the VHDL code as generic as possible. The only part of the VHDL code that isn't completely generic is the error check and assessor/corrector process as it was out of scope for this assignment.

This impulse response for the generic decoder when the number of error bits is set to 3 is:

<i>sin</i>	<i>rxlss(0)</i>	<i>rxlss(1)</i>	<i>rxlss(2)</i>	<i>sipo(6) & rxlss(0)</i>
1	0	0	0	1
0	1	0	0	0
0	0	1	0	1
0	1	0	1	1
0	1	1	0	1
0	1	1	1	0
0	0	1	1	0

From this impulse response, the Noise Look-Up table (**LUT**) can be given as:

<i>sin</i>	<i>sipo</i>
	6 5 4 3 2 1 0
0000001	0 0 1 1 1 0 1
0000010	0 1 1 1 0 1 0
0000100	1 1 1 0 1 0 0
0001000	1 1 0 1 0 0 0
0010000	1 0 1 0 0 0 0
0100000	0 1 0 0 0 0 0
1000000	1 0 0 0 0 0 0

- error code
- correction code
- ignored (data = "0000")


6.2. VHDL Code

```


1  library IEEE; -- include ieee library
2  use IEEE.STD_LOGIC_1164.all; -- include STD_LOGIC_1164 package from IEEE library to use the std_logic and
3  -- std_logic_vector data types which adds U (undefined) and Z (high impedance) assignments to the
4  -- standard VHDL bit and bit_vector data types
5  use IEEE.STD_LOGIC_UNSIGNED.all; -- Include STD_LOGIC_UNSIGNED package from IEEE library to be able to perform arithmetic,
6  -- conversion and comparison operations on the std_logic_vector data type
7
8  -- define the interface between the generic decoder and its external environment
9  ENTITY decoderGeneric IS
10     GENERIC (errorBits : NATURAL := 3); -- define the number of error bits e.g. if set to 3 then the decoder will be a 7/4
11     -- decoder
12     PORT (
13         clk, res, sin : IN STD_LOGIC; -- define the single point input ports (clock, reset & serial-in)
14         dout           : OUT STD_LOGIC_VECTOR(2**errorBits - (errorBits + 2) downto 0) -- define the output port (data-out) and
15         -- set it to be as wide as the number of
16         -- data bits
17     );
18 END decoderGeneric;
19
20 -- define the internal organisation and operation of the generic decoder
21 ARCHITECTURE rtl OF decoderGeneric IS
22     -- architecture declarations
23
24     -- calculate the number of data bits and total number of bits (error bits + data bits) given the number of error bits and
25     -- define them as constants
26     CONSTANT databits : NATURAL := 2**errorBits - (errorBits + 1);
27     CONSTANT allbits  : NATURAL := errorBits + databits;
28
29     SIGNAL rxlss : STD_LOGIC_VECTOR(errorBits - 1 downto 0) := (errorBits - 1 downto 0 => '0'); -- define the rxlss
30     -- (receive-linear-
31     -- sequential-system)
32     -- shift register and
33     -- set it to be as wide
34     -- as the number of
35     -- error bits
36     SIGNAL control : STD_LOGIC_VECTOR(allbits - 1 downto 0); -- define the control shift register and set it to be as
37     -- wide as the total number of bits
38     SIGNAL sipo    : STD_LOGIC_VECTOR(allbits - 1 downto 0) := (allbits - 1 downto 0 => '0'); -- define the sipo
39     -- (serial-in-parallel-
40     -- out) shift register
41     -- and set it to be
42     -- as wide as the total
43     -- number of bits
44
45     -- concurrent statements
46 BEGIN
47     -- define all linear sequential logic as a single process
48     PROCESS BEGIN
49         WAIT UNTIL RISING_EDGE (clk); -- ensures that each line of code in this process is dependent on a rising clock edge
50
51         -- when the reset port is set high, reset all registers to known state
52         IF (res = '1') THEN
53             control <= '1' & (allbits - 2 downto 0 => '0'); -- set the MSB high and all other bits low
54             rxlss <= (errorBits - 1 downto 0 => '0'); -- set all bits low
55             sipo <= (allbits - 1 downto 0 => '0'); -- set all bits low
56         -- when the reset port is set low, decode data coming in on the sin port sequentially and correct single bit errors
57         ELSIF (res = '0') THEN
58             -- decode the data coming in on the sin port by assigning the LSB of the rxlss register to the sin port XOR'ed
59             -- with the two MSB's of the rxlss register. However, when all bits have been received on the sin port (MSB of
60             -- the control register is low) set the LSB of the rxlss register low
61             rxlss(0) <= (sin XOR rxlss(errorBits - 2) XOR rxlss(errorBits - 1)) AND NOT(control(0));
62
63             -- shift the current value of each bit in the rxlss register up to the next MSB but set them low when all bits
64             -- have been received
65             FOR i IN 1 TO errorBits - 1 LOOP
66                 rxlss(i) <= rxlss(i - 1) AND NOT(control(0));
67             END LOOP;
68
69             -- store decoded data in sipo register by assigning the MSB to the sin port XOR'ed with the two MSB's of the
70             -- rxlss register and shifting the current value of each bit down to the next LSB
71             sipo <= (sin XOR rxlss(errorBits - 2) XOR rxlss(errorBits - 1)) & sipo(allbits - 1 downto 1);
72
73             -- keep track of the number of bits received sequentially on the sin port by shifting each bit of the control
74             -- register to the next LSB and looping the LSB back round to the MSB (a single bit will always be high while the
75             -- others are low)
76             control <= control(0) & control(allbits - 1 downto 1);
77
78             -- when the number of the error bits has been set to 3 and the sipo register has been filled (all 7-bits have been
79             -- received on sin port and decoded), compare the 3 MSB's of the sipo register (error bits) with the noise look-up
80             -- table for a 7/4 decoder
81             IF (control(allbits - 1) = '1') AND errorBits = 3 THEN
82                 CASE (sipo(allbits - 1 downto databits)) IS
83                     -- when single bit error is present XOR data bits of sipo with appropriate value from look-up table and send
84                     -- out the corrected data in parallel on the dout port
85                     WHEN "001" => dout <= (sipo(3 downto 0) XOR "1101");
86                     WHEN "011" => dout <= (sipo(3 downto 0) XOR "1010");
87                     WHEN "111" => dout <= (sipo(3 downto 0) XOR "0100");
88                     WHEN "110" => dout <= (sipo(3 downto 0) XOR "1000");
89                     -- when no single bit error is present or the data bits of sipo are "000" send out the data bits of sipo in
90                     -- parallel on the dout port
91                     WHEN OTHERS => dout <= sipo(3 downto 0);
92                 END CASE;
93
94             -- when the number of the error bits has been set to 4 and the sipo register has been filled (all 15-bits have been
95             -- received on sin port and decoded), compare the 4 MSB's of the sipo register (error bits) with the noise look-up
96             -- table for a 15/11 decoder
97             ELSIF (control(allbits - 1) = '1') AND errorBits = 4 THEN
98                 CASE (sipo(allbits - 1 downto databits)) IS
99                     -- when single bit error is present XOR data bits of sipo with appropriate value from look-up table and send
100                     -- out the corrected data in parallel on the dout port
101                     WHEN "0001" => dout <= (sipo(10 downto 0) XOR "1110111001");
102                     WHEN "0011" => dout <= (sipo(10 downto 0) XOR "11010110010");
103                     WHEN "0111" => dout <= (sipo(10 downto 0) XOR "10101100100");
104                     WHEN "1111" => dout <= (sipo(10 downto 0) XOR "01011001000");
105                     WHEN "1110" => dout <= (sipo(10 downto 0) XOR "10110010000");
106                     WHEN "1101" => dout <= (sipo(10 downto 0) XOR "01100100000");
107                     WHEN "1010" => dout <= (sipo(10 downto 0) XOR "11001000000");
108                     WHEN "1001" => dout <= (sipo(10 downto 0) XOR "10010000000");
109                     WHEN "0101" => dout <= (sipo(10 downto 0) XOR "00100000000");
110                     WHEN "0110" => dout <= (sipo(10 downto 0) XOR "01000000000");
111                     WHEN "1100" => dout <= (sipo(10 downto 0) XOR "10000000000");
112                     -- when no single bit error is present or the data bits of sipo are "000" send out the data bits of sipo in
113                     -- parallel on the dout port
114                     WHEN OTHERS => dout <= sipo(10 downto 0);
115                 END CASE;
116             END IF;
117         END PROCESS;
118     END rtl;

```


Compiled to schematic when errorBits = 3 (7/4 decoder)



Compiled to schematic when errorBits = 4 (15/11 decoder)





6.3. ModelSim Waveform



1. Reset registers to known state
2. Serially load in "0000001"
3. After 7-bits have been decoded (control = "1000000"), sipo = "0011101" which matches the impulse response on page 29

dout = "0000" throughout as sipo contains the error code "001" from the LUT which means the data bits from sipo (four LSB's) are XOR'ed with the corresponding correction code from the LUT

$$\begin{array}{lcl} \text{sipo} & \text{code} & \text{dout} \\ 1101 \oplus 1101 & = & 0000 \end{array}$$



3. Reset registers to known state
4. Serially load in "000000000000001"

4. After 15-bits have been decoded (control = “100000000000000”), sipo = “000111101011001” which matches the impulse response on page 12


dout = “00000000000” throughout as sipo contains the error code “0001” from the LUT which means the data bits from sipo (eleven LSB’s) are XOR’ed with the corresponding correction code from the LUT

$$\begin{array}{ccc} \textit{sipo} & \textit{code} & \textit{dout} \\ 11101011001 \oplus 11101011001 & = & 00000000000 \end{array}$$

6. Generic Encoder

7.1. Design

Key		
■	Entity	n Error Bits
■	Port	
■	Sequential Logic (dependent on clk)	
■	Register	
■	Combination Logic	



The diagram above illustrates the operation of the generic encoder which should be the same as that of the encoder described in **sections 3**. The difference being that the VHDL code for the generic encoder can be changed from a 7/4 to a 15/11 encoder by changing a single generic value. Again, the generic value that is altered is the number of error bits used. For a 7/4 encoder this would be 3 and for a 15/11 it would be 4. Error bits was chosen over other generic values as it ensures the coding system is perfect.

The Boolean expression for the logic which encodes the input signal is:

$$sout = sin \oplus rxlss(n - 2) \oplus rxlss(n - 1)$$

Where:

n = number of error bits

When the number of error bits is set to 3 the impulse response and LUT differs from that of the 7/4 encoder describe in **section 3** on **page 18** as the generic encoder always XOR's the two MSB's of rxlss rather than the MSB and LSB. This was done to keep the VHDL code for the encoder completely generic.

This impulse response for the generic encoder when the number of error bits is set to 3 is:

<i>sin</i>	<i>rxlss(0)</i>	<i>rxlss(1)</i>	<i>rxlss(2)</i>	<i>sout</i>
1	0	0	0	1
0	1	0	0	0
0	0	1	0	1
0	0	0	1	1
0	0	0	0	0
0	0	0	0	0
0	0	0	0	0

From this impulse response, the Noise Look-Up table (**LUT**) can be given as:

<i>sin</i>	<i>sout</i>
0000001	0001101
0000010	0011010
0000100	0110100
0001000	1101000
0010000	1010000
0100000	0100000
1000000	1000000

The impulse response for the generic encoder when the number of error bits is set to 4 is:

<i>sin</i>	<i>rxlss(0)</i>	<i>rxlss(1)</i>	<i>rxlss(2)</i>	<i>rxlss(3)</i>	<i>sout</i>
1	0	0	0	0	1
0	1	0	0	0	0
0	0	1	0	0	0
0	0	0	1	0	1
0	0	0	0	1	1
0	0	0	0	0	0
0	0	0	0	0	0
0	0	0	0	0	0
0	0	0	0	0	0
0	0	0	0	0	0
0	0	0	0	0	0
0	0	0	0	0	0
0	0	0	0	0	0
0	0	0	0	0	0
0	0	0	0	0	0

From this impulse response, the Noise Look-Up table (**LUT**) can be given as:

<i>sin</i>	<i>sout</i>
0000000000000001	000000000011001
0000000000000010	000000000110010
00000000000000100	0000000001100100
000000000000001000	000000011001000
0000000000000010000	000000110010000
00000000000000100000	000001100100000
000000000000001000000	000011001000000
0000000000000010000000	000110010000000
00000000000000100000000	001100100000000
000000000000001000000000	011001000000000
0000000000000010000000000	110010000000000
0001000000000000	100100000000000
00100000000000000	001000000000000
010000000000000000	010000000000000
1000000000000000000	100000000000000


7.2. VHDL Code

```


1  library IEEE; -- include ieee library
2  use IEEE.STD_LOGIC_1164.all; -- include STD_LOGIC_1164 package from IEEE library to use the std_logic and
3  -- std_logic_vector data types which adds U (undefined) and Z (high impedance) assignments to the
4  -- standard VHDL bit and bit_vector data types
5  use IEEE.STD_LOGIC_UNSIGNED.all; -- Include STD_LOGIC_UNSIGNED package from IEEE library to be able to perform arithmetic,
6  -- conversion and comparison operations on the std_logic_vector data type
7
8  -- define the interface between the generic encoder and its external environment
9  ENTITY encoderGeneric IS
10     GENERIC (errorBits : NATURAL := 3); -- define the number of error bits e.g. if set to 3 then the encoder will be a 7/4
11     -- encoder
12     PORT (
13         clk, res, sin : IN STD_LOGIC; -- define the single point input ports (clock, reset & serial-in)
14         sout           : OUT STD_LOGIC -- define the single point output port (serial-out)
15     );
16 END encoderGeneric;
17
18 -- define the internal organisation and operation of the generic encoder
19 ARCHITECTURE rtl OF encoderGeneric IS
20     -- architecture declarations
21     SIGNAL rxlss : STD_LOGIC_VECTOR(0 to errorBits - 1) := (others => '0'); -- define the rxlss (receive-linear-sequential-
22     -- system) shift register and set it to be as
23     -- wide as the number of error bits
24
25     -- concurrent statements
26 BEGIN
27     -- combination logic to send encoded data out on the sout port by XOR'ing the sin port with the two MSB's of the rxlss
28     -- register
29     sout <= sin XOR rxlss(errorBits - 2) XOR rxlss(errorBits - 1);
30
31     -- define all linear sequential logic as a single process
32     PROCESS BEGIN
33         WAIT UNTIL RISING_EDGE (clk); -- ensures that each line of code in this process is dependent on a rising clock edge
34
35         -- when the reset port is set high, reset register to known state
36         IF (res = '1') THEN
37             rxlss <= (others => '0'); -- set all bits low
38         -- when the reset port is set low, assign the LSB of the rxlss register to the sin port and shift each bit of the
39         -- rxlss register up to the next MSB
40         ELSIF (res = '0') THEN
41             rxlss <= sin & rxlss(0 to errorBits - 2);
42         END IF;
43     END PROCESS;
44 END rtl;

```


Compiled to schematic when errorBits = 3 (7/4 encoder)



Compiled to schematic when errorBits = 4 (15/11 encoder)





7.3. ModelSim Waveform



1. Reset registers to known state
2. Serially load in "0000001" and send encoded signal out


Due to the delay in the combination logic, the decoder will read falling edges on sout as "1" and rising edges as "0" hence sout matches the impulse response "0001101" seen on **page 17**






1. Reset registers to known state
2. Serially load in "0000000000000001" and send encoded signal out

Due to the delay in the combination logic, the decoder will read falling edges on sout as "1" and rising edges as "0" hence sout matches the impulse response "000000000011001" seen on **page 17**



7. Generic Linear Sequential Coding System

8.1. Design



The diagram above illustrates the operation of the generic LSS which should be the same as that of the LSS described in **sections 4**. The difference being that the VHDL code for the generic encoder can be changed from a 7/4 to a 15/11 LSS by changing a single generic value. Again, the generic value that is altered is the number of error bits used. For a 7/4 LSS this would be 3 and for a 15/11 it would be 4. Error bits was chosen over other generic values as it ensures the coding system is perfect.


8.2. VHDL Code

```


1  library IEEE; -- include ieee library
2  use IEEE.STD_LOGIC_1164.all; -- include STD_LOGIC_1164 package from IEEE library to use the std_logic and
3  -- std_logic_vector data types which adds U (undefined) and Z (high impedance) assignments to the
4  -- standard VHDL bit and bit_vector data types
5  use IEEE.STD_LOGIC_UNSIGNED.all; -- Include STD_LOGIC_UNSIGNED package from IEEE library to be able to perform arithmetic,
6  -- conversion and comparison operations on the std_logic_vector data type
7
8  -- define the interface between the generic linear sequential coding system (lss) and its external environment
9  ENTITY lssGeneric IS
10     GENERIC (errorBits : NATURAL := 4); -- define the number of error bits e.g. if set to 3 then the lss will be a 7/4
11     PORT (
12         clk, res, test, sin : IN STD_LOGIC; -- define the single point input ports (clock, reset, serial-in & test)
13         dout : OUT STD_LOGIC_VECTOR(2*errorBits - (errorBits + 2) downto 0) -- define the output port (data-out) and set it to
14         -- be as wide as the number of data bits
15     );
16 END lssGeneric;
17
18 -- define the internal organisation and operation of the generic lss
19 ARCHITECTURE rtl OF lssGeneric IS
20     -- architecture declarations
21     SIGNAL t1 : STD_LOGIC; -- define the serial transmission line between the output of the encoder and input of the decoder
22
23     -- concurrent statements
24 BEGIN
25     -- combination logic to instantiate the generic encoder and generic decoder entities inside the generic lss
26     encoderGeneric : ENTITY work.encoderGeneric(rtl)
27     GENERIC MAP (
28         errorBits => errorBits -- pass the number of error bits set in the generic lss to the generic encoder
29     )
30     PORT MAP (
31         sin => sin, -- connect the sin port of the generic encoder directly to the sin port of the generic lss
32         clk => clk, -- connect the clk port of the generic encoder directly to the clk port of the generic lss
33         res => res, -- connect the res port of the generic encoder directly to the res port of the generic lss
34         sout => t1 -- connect the sin port of the generic encoder to the transmission line
35     );
36
37     decoderGeneric : ENTITY work.decoderGeneric(rtl)
38     GENERIC MAP (
39         errorBits => errorBits -- pass the number of error bits set in the generic lss to the generic decoder
40     )
41     PORT MAP (
42         sin => t1 XOR test, -- connect the sin port of the generic decoder to the transmission line XOR'ed with the test
43         -- port of the generic lss. This allows errors to be introduced as the signal coming out of
44         -- the generic encoder will be inverted when the test port is set high
45         clk => clk, -- connect the clk port of the generic decoder directly to the clk port of the generic lss
46         res => res, -- connect the res port of the generic decoder directly to the res port of the generic lss
47         dout => dout -- connect the dout port of the generic decoder directly to the dout port of the generic lss
48     );
49 END rtl;

```


Compiled to schematic when errorBits = 3 (7/4 LSS)



Compiled to schematic when errorBits = 4 (15/11 LSS)




8.3. ModelSim Waveform



1. Reset registers to known state
2. Serially load in "0000110" and introduce single bit error
3. Serially load in "0000001" and introduce single bit error

`dout` = 0110 which matches the data bits (first 4 bits) of the 7-bit message load in during **step 2** hence the single bit error was corrected successfully

4. `dout` = 0001 which matches the data bits of the 7-bit message load in during **step 3** hence the single bit error was corrected successfully



1. Reset registers to known state
2. Serially load in "000011000111001" and introduce single bit error
3. Serially load in "110000000000000" and introduce single bit error


dout = 11000111001 which matches the data bits (first 11 bits) of the 15-bit message load in during **step 2** hence the single bit error was corrected successfully

4. dout = 00000000001 which matches the data bits of the 15-bit message load in during **step 3** hence the single bit error was corrected successfully

8. Parallel-In Encoder

9.1. Design

Key			
■	Entity	n	Error Bits
■	Port	d	Data Bits ($2^{n-(n+1)}$)
■	Sequential Logic (dependent on clk)	a	All Bits ($n + 2^{n-(n+1)}$)
■	Register		
■	Combination Logic		



One problem with the encoder used in all the previous sections is that the entire 7-bit message needs to be loaded in serially. This means the three error bits (last three bits) could be set to “1” which would prevent the message from being correctly decoded. A solution to this problem is to load the data in parallel. The diagram above illustrates the operation of the parallel 7/4 encoder which should be:

1. Can be reset at any time by setting the res port high
2. Decodes messages sent in, in parallel on the din port
3. Outputs the encoded messages serially
4. Continues to operate once the message has been encoded i.e. once the first message has been encoded, any new data on din will be read automatically

The purpose of each component illustrated in the diagram above is:

- **din input port** – receives input message
- **clk input port** - receives clock signal
- **res input port** - receives reset signal
- **sout output port** – transmits encoded signal
- **⊕ logic** – encodes input signal
- **rxlss shift register** – stores the last n bits input from the data register to encode the message

- **data shift register** – stores the input message
- **control shift register** – keeps track of the number of bits encoded. When MSB is “1” it loads signals on din into the data register

The Boolean expression for the logic which encodes the input signal is:

$$sout = data(d - 1) \oplus rxlss(n - 2) \oplus rxlss(n - 1)$$

Where:

$n = \text{number of error bits}$

$d = \text{number of data bits } (2^{n-(n+1)})$

The impulse response for the generic parallel encoder when the number of error bits is set to 3 is:

<i>data(3)</i>	<i>rxlss(0)</i>	<i>rxlss(1)</i>	<i>rxlss(2)</i>	<i>sout</i>
1	0	0	0	1
0	1	0	0	0
0	0	1	0	1
0	0	0	1	1
0	0	0	0	0
0	0	0	0	0
0	0	0	0	0

From this impulse response, the Noise Look-Up table (**LUT**) can be given as:

<i>din</i> 0 1 2 3	<i>sout</i>
0 0 0 1	0001101
0 0 1 0	0011010
0 1 0 0	0110100
1 0 0 0	1101000


9.2. VHDL Code

```


1  library IEEE; -- include ieee library
2  use IEEE.STD_LOGIC_1164.all; -- include STD_LOGIC_1164 package from IEEE library to use the std_logic and
3  -- std_logic_vector data types which adds U (undefined) and Z (high impedance) assignments to the
4  -- standard VHDL bit and bit_vector data types
5  use IEEE.STD_LOGIC_UNSIGNED.all; -- Include STD_LOGIC_UNSIGNED package from IEEE library to be able to perform arithmetic,
6  -- conversion and comparison operations on the std_logic_vector data type
7
8  -- define the interface between the generic encoder and its external environment
9  ENTITY pisoEncoderGeneric IS
10     GENERIC (errorBits : NATURAL := 3); -- define the number of error bits e.g. if set to 3 then the encoder will be a 7/4
11     -- encoder
12     PORT (
13         clk, res : IN STD_LOGIC; -- define the single point input ports (clock & reset)
14         pin      : IN STD_LOGIC_VECTOR(2*errorBits - (errorBits + 2) downto 0); -- define the input port (parallel-in) and
15         -- set it to be as wide as the number of
16         -- data bits
17         sout     : OUT STD_LOGIC -- define the single point output port (serial-out)
18     );
19 END pisoEncoderGeneric;
20
21 -- define the internal organisation and operation of the generic encoder
22 ARCHITECTURE rtl OF pisoEncoderGeneric IS
23     -- architecture declarations
24
25     -- calculate the number of data bits and total number of bits (error bits + data bits) given the number of error bits and
26     -- define them as constants
27     CONSTANT databits : NATURAL := 2*errorBits - (errorBits + 1);
28     CONSTANT allBits  : NATURAL := errorBits + databits;
29
30     SIGNAL data      : STD_LOGIC_VECTOR(databits - 1 downto 0) := (others => '0'); -- define the data shift register and set
31     -- it to be as wide as the number of data
32     -- bits
33     SIGNAL control   : STD_LOGIC_VECTOR(allBits - 1 downto 0); -- define the control shift register and set it to be as
34     -- wide as the total number of bits
35     SIGNAL rxlss     : STD_LOGIC_VECTOR(0 to errorBits - 1) := (others => '0'); -- define the rxlss (receive-linear-
36     -- sequential-system) shift register and
37     -- set it to be as wide as the number of
38     -- error bits
39
40     -- concurrent statements
41 BEGIN
42     -- combination logic to send encoded data out on the sout port by XOR'ing the sin port with the two MSB's of the rxlss
43     -- register
44     sout <= data(databits - 1) XOR rxlss(errorBits - 2) XOR rxlss(errorBits - 1);
45
46     -- define all linear sequential logic as a single process
47     PROCESS BEGIN
48         WAIT UNTIL RISING_EDGE (clk); -- ensures that each line of code in this process is dependent on a rising clock edge
49
50         -- when the reset port is set high, reset register to known state
51         IF (res = '1') THEN
52             control <= '1' & (allBits - 2 downto 0 => '0'); -- set the MSB high and all other bits low
53             rxlss <= (others => '0'); -- set all bits low
54             data <= (others => '0'); -- set all bits low
55             -- when the reset port is set low, shift data out of the data register serially and encode it
56             ELSIF (res = '0') THEN
57                 -- assign the LSB of the rxlss register to the MSB of the data register and shift each bit of the rxlss register
58                 -- up to the next MSB
59                 rxlss(0) <= data(databits - 1);
60                 FOR i IN 1 TO errorBits - 1 LOOP
61                     rxlss(i) <= rxlss(i - 1);
62                 END LOOP;
63
64                 -- shift each bit of the data register up to the next MSB
65                 data <= data(databits - 2 downto 0) & '0';
66
67                 -- keep track of the number of bits shifted out of the data register sequentially by shifting each bit to the
68                 -- next LSB and looping the LSB back round to the MSB (a single bit will always be high while the others are low)
69                 control <= control(0) & control(allBits - 1 downto 1);
70
71                 -- when the encoder is reset or all bits of the previous data set have been encoded (MSB of the control register
72                 -- is high), load data on the pin port into the data register in parallel
73                 IF (control(allBits - 1) = '1') THEN
74                     FOR i IN 0 TO databits - 1 LOOP
75                         data(i) <= pin(i);
76                     END LOOP;
77                 END IF;
78             END IF;
79         END PROCESS;
80     END rtl;

```

Compiled to schematic



9.3. ModelSim Waveform



1. Reset registers to known state
2. Load "0001" in parallel and send encoded signal out

`sout` matches that of the impulse response on **page 48** but unlike the other encoders, it is now bound by the clock as the input is loaded into a shift register that use sequential logic